
Abstract—Since plasticity is path dependent, it is necessary to
properly take into account the deformation, strain rate and
temperature history in applications such as crash worthiness
and ballistics simulations. To accurately model the evolution
of the yield stress, the incremental (differential) update from
a previous converged time step is required instead of a closed
form expression that relates flow stress to plastic strain. Elasto-
viscoplastic models that make use of state variables better capture
the physical phenomenon of a perceived lag between a change in
strain rate or temperature and the subsequent stress response. It
is impossible to capture this when making use of a closed form
expression or data table based method. One model that makes use
of an evolving state variable is the Mechanical Threshold Stress
(MTS) model. In this paper, the implementation of the MTS
model into an Abaqus user hardening (UHARD) subroutine is
discussed and the code is included. The aim is not to improve on
the current knowledge of the model, but to illustrate the ease with
which a state variable based plasticity model can be implemented
and used instead of an empirical (closed form expression) or
data table based method. The MTS model is compared to the
Johnson-Cook plasticity model which takes the form of a simple
closed form expression relating yield stress to plastic strain as a
function of temperature and strain rate. The model parameters
are calibrated using isothermal, constant strain rate experimental
data and then used to predict the stress response for a strain rate
jump test and a temperature change test.

Keywords—Mechanical Threshold Stress (MTS); Plasticity;
Abaqus User Subroutines; UHARD

INTRODUCTION

The Mechanical Threshold Stress (MTS) model [2] was
developed to describe the post yielding behaviour of metals.
It has demonstrated the ability to accurately model the effect
of temperature and plastic strain rate on the post yielding
behaviour of metals.

In this paper, the MTS model is calibrated using isothermal,
constant strain rate data on OFHC Cu digitised from the Ph.D.
thesis by A.B. Tanner [6] as an example. The model is then
used to predict a strain rate jump test and temperature change
test. Data for these experimental tests are also available in
[6]. The Johnson-Cook plasticity model [3] is calibrated on
the same data and also used to predict the strain rate jump
and temparature change response for comparison.

This paper provides a tutorial on the MTS model and its
implementation into an Abaqus environment. The code of the
Abaqus user subroutine is included at the end of the article.

I. THEORY

In the MTS model, a material has a theoretical maximum
flow stress at 0 K, called the mechanical threshold, σ̂. The
material flow stress, σy , is obtained by scaling the mechanical
threshold to accommodate rate and temperature dependence.

Firstly, the threshold flow stress can be separated into an
athermal component σ̂a and thermal components σ̂κt .

σ̂ = σ̂a +
∑
κ

σ̂κt . (1)

The athermal component σ̂a characterises the rate-independent
interactions of dislocations with long-range barriers. The ther-
mal components σ̂κt characterise the rate-dependent interac-
tions of dislocations with short-range obstacles that can be
overcome with the assistance of thermal activation [2].

At different temperatures T and plastic strain rates ε̇, the
contributions to the flow stress σκt are related to their threshold
counterparts σ̂κt through the scaling functions Sκt (ε̇, T ), so that

σκt = σ̂κt S
κ
t (ε̇, T ). (2)

Considering the scaling relationship between the flow stress
and theoretical threshold values of a material, σy can be
expressed as

σy
µ

=
σ̂a
µ

+
∑
κ

σκt
µ

=
σ̂a
µ

+
∑
κ

Sκt (ε̇, T )
σ̂κt
µo
. (3)

Here, µo is a reference value of the shear modulus µ, which
is often modeled by [7]

µ = µ̃(T ) = µo −
Do

exp
(
To
T

)
− 1

, (4)

in which To and Do are empirical constants. The temperature
dependence of µ is included in the scaling functions Sκt .
The interaction kinetics for short-range obstacles are described
using an Arrhenius expression while a phenomenological
relation is used for the free energy function of stress [4].

The scaling functions Sκt (ε̇, T ) now take the form

Sκt (ε̇, T ) =

[
1−

(
kT

gκotµb
3
ln
ε̇κot
ε̇

)1/qκt
]1/pκt

, (5)

where k is the Boltzmann constant, b is the magnitude of
the Burger’s vector, go is the normalized activation energy for
dislocations to overcome the obstacles, ε̇o is a constant and

p and q are statistical constants that characterize the shape of
the obstacle profile (0 ≤ p ≤ 1 , 1 ≤ q ≤ 2) [4].

In the standard MTS model there are two thermal com-
ponents, i.e. σ̂κt , κ = 1, 2. Using the notation σ̂1

t = σ̂i and
σ̂2
t = σ̂ε, the flow stress relation in (3) changes to

σy
µ

=
σ̂a
µ

+ Si(ε̇, T )
σ̂i
µo

+ Sε(ε̇, T )
σ̂ε
µo
, (6)

In (6), σ̂i describes the thermal portion of the yield stress
(non-evolving thermal stress component) and σ̂ε describes the
interaction of mobile dislocations with the forest dislocation
structure (evolving component).

The evolution of σ̂ε is given in rate form, by

dσ̂ε
dε

= θ(T, ε̇, σ̂ε) = θo − θr(T, ε̇, σ̂ε), (7)

where θo is the hardening due to dislocation accumulation
(assumed constant) and θr is the dynamic recovery rate.
The functional form of the hardening rate θ is chosen to fit
experimental data. The tanh functional form [1], [2]

θ = θo

1−
tanh

[
ασ̂ε
σ̂εs

]
tanh(α)

 (8)

is used here although the power law form [5] is also a popular
choice. The α parameter is a fitted constant and σ̂εs is the
saturation threshold stress. θo assumes the role of the initial
hardening rate. The hardening rate θ decreases with strain
and reaches saturation. The saturation threshold stress σ̂εs is
a function of both strain rate and temperature, through [1]

ln
ε̇

ε̇εso
=
goεsµb

3

kT
ln

σ̂εs
σ̂εso

(9)

where ε̇εso, goεs and σ̂εso are empirically obtained constants.

II. IMPLEMENTATION INTO ABAQUS

The MTS material model is implemented into an Abaqus
UHARD user subroutine. This user subroutine is included in
Section IV. The included subroutine can be used to calculate
the yield stress (SYIELD) and gradients of yield stress with
respect to equivalent plastic strain (HARD(1)) and equivalent
plastic strain rate (HARD(2)). The UHARD subroutine only
takes care of the plastic response and the user should still
account for the temperature dependence of the elastic response.

A limitation in the implemented subroutine is that the
mechanical threshold scaling functions in (5) are calculated
for a plastic strain rate above 10−8. The subroutine can
therefore not distinguish between plastic strain rates below this
arbitrarily selected threshold value. The gradients with respect
to equivalent plastic strain and equivalent plastic strain rate
(HARD(1) = ∂σy

∂εp
and HARD(2) = ∂σy

∂ε̇p
) are evaluated analytically

to allow quadratic convergence.
The MTS UHARD user subroutine requires 19 material

parameters and 8 state dependent variables. The definition of
these material parameters and state dependent variables are
given in Table I. In the subsections to follow, the paper will

cover the calculation of the evolving thermal stress component
and analytical gradients as well as how and why the state
dependent variables are used.

TABLE I
ABAQUS AND MTS UHARD SUBROUTINE MATERIAL PROPERTY

DEFINITION.

Variable Equations UHARD ABAQUS
µo (4) ZMU0 PROPS(1)
Do (4) D0 PROPS(2)
To (4) TEMP0 PROPS(3)
σ̂a (1), (3), (6) SA PROPS(4)
σ̂i (6) SI PROPS(5)
σ̂ε (6), (7), (8) SE0 PROPS(6)
σ̂εso (9) SES0 PROPS(7)
θo (7), (8) TH0 PROPS(8)
k/b3 (5), (9) ZKB3 PROPS(9)
α (8) ALPHA PROPS(10)
goi (5) G0I PROPS(11)
goε (5) G0E PROPS(12)
goεs (9) G0ES PROPS(13)
ε̇o (5) ER0 PROPS(14)
ε̇εso (9) ER0ES PROPS(15)
qi (5) QI PROPS(16)
pi (5) PI PROPS(17)
qε (5) QE PROPS(18)
pε (5) PE PROPS(19)

A. Calculating the evolving thermal component

Midpoint integration is used to determine the value of
the evolving thermal stress component (σ̂ε) at the end of a
specific time step. The plastic strain rate, duration of the time
increment and change in temperature over the time increment
are taken into account. The value of the evolving thermal stress
component at time tn+1 is now given by

σ̂n+1
ε = σ̂nε +

4εp
2

(
θ(Tn, ε̇n, σ̂nε ) + θ(Tn+1, ε̇n+1, σ̂n+1

ε )
)

= σ̂nε +
4εp
2

(
θn + θn+1

)
.

(10)

The choice of the tanh functional form and midpoint integra-
tion require solution by the Newton-Raphson method. In the
Abaqus UHARD implementation, an initial guess is assigned
σ̂n+1
ε = σ̂nε . The update of σ̂n+1

ε is then performed iteratively
by the Newton-Raphson method:

(
σ̂n+1
ε

)
i+1

=
(
σ̂n+1
ε

)
i
−
(
f(σ̂n+1

ε )
)
i(

f ′(σ̂n+1
ε )

)
i

. (11)

Equation (11) is iterated to convergence for a fixed timestep
at a specific element integration point, where

f(σ̂n+1
ε ) = σ̂n+1

ε − σ̂nε −
4εp
2

(
θn + θn+1

)
and

f ′(σ̂n+1
ε ) = 1− 4εp

2

dθn+1

dσ̂n+1
ε

.

For the tanh functional form in (8),

θn+1 = θo

1− tanh
(
ασ̂n+1

ε

σ̂n+1
εs

)
tanh(α)

 , (12)

and hence

dθn+1

dσ̂n+1
ε

=
−αθo

σ̂n+1
εs tanh(α) cosh2

(
ασ̂n+1

ε

σ̂n+1
εs

) . (13)

B. Analytical gradients

The partial derivatives HARD(1) and HARD(2) are evaluated
analytically to allow quadratic convergence of the displace-
ment solution. These gradients represent the change in the
calculated SYIELD for a different EQPLAS and EQPLASRT input
respectively.

Using the general MTS definition in (6), the yield stress at
the end of the current increment, σn+1

y , is

σn+1
y = σ̂a +

µn+1

µo

(
Sn+1
i σ̂i + Sn+1

ε σ̂n+1
ε

)
. (14)

The first partial derivative with respect to EQPLAS = εn+1
p is

determined analytically by

HARD(1) =
∂σn+1

y

∂εn+1
p

=
µn+1

µo

(
Sn+1
ε

dσ̂n+1
ε

dεn+1
p

)
. (15)

Given the midpoint evaluation of σ̂n+1
ε in (10), the tanh func-

tional form of the hardening rate from (12) and an equivalent
plastic strain update 4εp = εn+1

p − εnp :

dσ̂n+1
ε

dεn+1
p

=
θn + θn+1

2
+
4εp
2

dθn+1

dεn+1
p

. (16)

Now, using (13) and the chain rule

dθn+1

dεn+1
p

=
dθn+1

dσ̂n+1
ε

dσ̂n+1
ε

dεn+1
p

, (17)

the partial derivative with respect to the equivalent plastic
strain HARD(1) is determined by

HARD(1) =
∂σn+1

y

∂εn+1
p

=
µn+1Sn+1

ε (θn + θn+1)

2µo(1−A)
; (18)

A =
4εp
2

dθn+1

dσ̂n+1
ε

.

Similarly, the second partial derivative with respect to
EQPLASRT = ε̇n+1

p is determined from

HARD(2) =
∂σn+1

y

∂ε̇n+1
p

=
µn+1

µo

(
dSn+1

i

dε̇n+1
p

σ̂i +
dSn+1

ε

dε̇n+1
p

σ̂n+1
ε + Sn+1

ε

dσ̂n+1
ε

dε̇n+1
p

)
.

(19)

Using the Arrhenius expression for the scaling functions in
(5), the partial derivatives of the scaling functions with respect
to the equivalent plastic strain rate can be determined from

d(Sκt )
n+1

dε̇n+1
p

=
γκt

pκt q
κ
t ε̇
n+1
p

[
1−

(
γκt ln

(
ε̇κot
ε̇n+1
p

)) 1
qκ
t

] 1−pκ
t

pκ
t

×
(
γκt ln

(
ε̇κot
ε̇n+1
p

)) 1−qκ
t

qκ
t

; (20)

γκt =
kTn+1

gκotµ
n+1b3

.

Using (9), the midpoint evaluation of σ̂n+1
ε in (10), the chain

rule
dθn+1

dε̇n+1
p

=
dθn+1

dσ̂n+1
ε

dσ̂n+1
ε

dε̇n+1
p

(21)

and the tanh functional form of the hardening rate from (12),

dσ̂n+1
ε

dε̇n+1
p

=
A

(A− 1)

γεsσ̂εsoσ̂
n+1
ε

ε̇oεsσ̂
n+1
εs

(
ε̇n+1
p

ε̇oεs

)γεs−1
; (22)

γεs =
kTn+1

goεsµn+1b3
; A =

4εp
2

dθn+1

dσ̂n+1
ε

.

In order to evaluate HARD(2) analytically, (20) and (22) are
determined first and then simply substituted into (19).

C. State dependent variables

The implemented user subroutine makes use of 8 state
dependent variables. The first state dependent variable,
STATEV(1) stores the value of the evolving thermal stress
component σ̂nε at the end of the previous increment. STATEV(2)
stores the current update in the evolving thermal stress com-
ponent, 4σ̂n+1

ε . In order to evaluate θn and so properly use
midpoint integration in (10) for example, the equivalent plastic
strain as well as the equivalent plastic strain rate at the end of

TABLE II
MTS UHARD STATE DEPENDENT VARIABLE DEFINITION AND

ALLOCATION.

Variable Definition Allocation
σ̂n
ε σ̂ε at time tn, the start of the current STATEV(1)

increment.
4σ̂n+1

ε Change in σ̂ε for current increment. STATEV(2)
εpn Equivalent plastic strain εp at the end STATEV(3)

of the previous increment.
εpn+1 Equivalent plastic strain εp at the end STATEV(4)

of the current increment.
ε̇pn Equivalent plastic strain rate ε̇p at the STATEV(5)

end of the previous increment.
ε̇pn+1 Equivalent plastic strain rate εp at the STATEV(6)

end of the current increment.
KINC KINC verification counter allowing a STATEV(7)

single update of the relevant variables
KSTEP KSTEP verification counter used to STATEV(8)

reset STATEV(7) should the analysis
move on to a different step.

Fig. 1. Flow chart of Abaqus process indicating state variables used in the UHARD user subroutine. After an increment has converged, values of the state
dependent variables STATEV(2), STATEV(4) and STATEV(6) can be updated to state dependent variables STATEV(1), STATEV(3) and STATEV(5)
for use in the next solution increment.

the previous increment is required. STATEV(3) and STATEV(5)

are used to store these values of εnp and ε̇np .
Since multiple calls to the user hardening subroutine are

required for convergence, it is imperative that the evolving
thermal stress component as well as the equivalent plastic
strain and equivalent plastic strain rate are only stored in
STATEV(1), STATEV(3) and STATEV(5) for use in the next
increment once a converged solution is obtained. The state
dependent variable STATEV(7) is used to ensure that these state
dependent variables are only updated for the first UHARD
subroutine evaluation in a specific increment. These updates
therefore take place when the yield condition is checked using
the state dependent variable values at the end of the previous
converged solution.

The state dependent variable updates are as follow:

• The converged value of the evolving thermal stress com-
ponent update 4σ̂ε for a previous increment stored in
STATEV(2) is added to STATEV(1).

• STATEV(3) is updated to contain the converged value of
EQPLAS for the previous increment stored in STATEV(4).

• STATEV(5) is updated to contain the converged value of
EQPLASRT for the previous increment stored in STATEV(6).

STATEV(8) is used to reset the increment counter STATEV(7)

to 0 if the simulation moves on to another step. The definition
and allocation of the state dependent variables is also given in
Table II, with a flow chart of the Abaqus solution procedure
and state dependent variable updates visible in Figure 1.

III. EXAMPLES

As an example of the MTS model performance, the model
parameters are calibrated using isothermal, constant strain rate
data on OFHC Cu digitised from the Ph.D. thesis by A.B. Tan-
ner [6]. The Johnson-Cook plasticity model [3] is calibrated
on the same data. In the Johnson-Cook plasticity model, the

Fig. 2. Isothermal constant strain rate stress-strain data for OFHC Copper
in compression as digitised from [6].

flow stress is given by the closed form expression [3]

σy =
(
σ0 +Bεnp

)(
1 + C ln

ε̇p
ε̇0

)(
1−

[
T − Tr
Tm − Tr

]m)
.

(23)

The data digitised from the Ph.D. of A.B. Tanner is visible
in Figure 2. These stress-strain curves are of compression data
on OFHC copper at:
• a temperature of 25oC and strain rate of 0.0004s−1;
• a temperature of 269oC and strain rate of 0.1s−1 and
• a temperature of 269oC and strain rate of 0.0004s−1.

The model implemented and discussed in this article is inca-
pable of describing recrystalisation or material softening. For
this reason, the stress-strain data at 269oC and strain rate of
0.0004s−1 was only used up to a true strain of approximately
55%.

The elastic properties of the material was chosen as that
descibed in [6]. For consistency, both MTS and Johnson-Cook
models use the shear modulus relation

µ = 1000
[
47.093−

(
0.1429 + 0.0002763T 2

)0.5]
(24)

(a) (b)

Fig. 3. Tuned (a) Johnson-Cook and (b) Mechanical Threshold Stress material models for isothermal, constant strain rate compression data.

(a) (b) (c) (d)

Fig. 4. Comparison of Johson-Cook and MTS calculated response compared to experimental data of a strain rate jump test and temperature change test after
50% compression. (a) Johnson-Cook and (b) MTS result compared to the strain rate jump test data of [6]. (c) Johnson-Cook and (d) MTS result compared
to the temperature change test data of [6].

instead of the relationship in (4). Using ν = 1/3 [6], the elastic
modulus is obtained from the shear modulus by using

E = 2µ(1 + ν). (25)

An initial estimate on the model parameter values were
obtained in [6]. An optimisation routine is then run to calibrate
the relevant parameters by comparing the numerical stress-
strain response to the isothermal, constant strain rate data
visible in Figure 2. For the MTS model calibration, the
parameters σ̂a, σ̂i, σ̂εso, θo, α, goi, goε and goεs were tuned
using this optimisation procedure. The Johnson-Cook model
parameters σ0, B, n, C and m were tuned in the same way.

The material model parameter values for both plasticity
models are visible in Table III. The curve fits that result in
the best match with the experimental data is displayed in
Figure 3(a) for the Johnson-Cook model and in Figure 3(b)
for the Mechanical Threshold stress model. In these figures
it is visible that the physically based MTS model (initially
developed specifically for Copper) is better suited to replicate
the isothermal constant strain rate OFHC Cu data.

Two additional experimental data sets were also digitised
from [6]. In the first experiment, the specimen was kept at
269oC. It is first compressed up to 50% strain at a constant
strain rate of 0.0004s−1 and then at a constant strain rate
of 0.1s−1. Using the Johnson-Cook and MTS models with

the material model parameter values in Table III and elastic
response in (24) and (25), the experimental strain and strain
rate history is fed into the material models resulting in
Figure 4(a) for the Johnson-Cook model and Figure 4(b) for
the MTS model respectively.

TABLE III
MTS AND JC MATERIAL PROPERTY VALUES CALIBRATED ON DATA.

MTS JC
Variable Value Variable Value
To 208 σ0 0.0104
σ̂a 1.635 B 419.67
σ̂i 0.324 n 0.364
σ̂ε 0 C 0.0195
σ̂εso 412.12 ε̇0 0.0001
θo 2026.83 Tr 208
k/b3 0.848 Tm 1070
α 1.799 m 0.91
goi 0.027
goε 1.313
goεs 0.553
ε̇o 1e7

ε̇εso 1e7
qi 1.5
pi 0.5
qε 1
pε 0.6667

(a) (b)

Fig. 5. Comparison of (a) Table based method and (b) Abaqus MTS UHARD
subroutine for a strain rate jump test.

In the second experiment, the test was carried out at a
constant strain rate of 0.0004s−1. The specimen is kept at
269oC up to 50% strain and then cooled down to 25oC before
compressing the specimen further. The result of the Johnson-
Cook model is visible in Figure 4(c) with the result of the
MTS model visible in Figure 4(d).

The Johnson-Cook model consists of a closed form ex-
pression and the change in strain rate or temperature simply
translates into a jump from one curve to another. This simple
jump between curves would be true for any model that
describes the relationship between stress and strain subject
to strain rate and temperature dependence in a closed form
expression.

Figure 5 illustrates the difference in the results of a strain
rate jump test example if done using the UHARD and table
based methods in Abaqus. This is to illustrate what happens
when using the table based method in Abaqus versus the user
hardening subroutine.

Given a selection of MTS material parameters and strain rate
dependent tables, the black lines in Figures 5(a) and (b) are
recovered if a single element is compressed 10% in a 1000s
and 0.1s period respectively. These curves are the reference
curves for a strain rate of 0.0001s−1 and 1s−1. In the strain
rate jump test, the element is first compressed 4.5% in a 400s
period, it is then compressed an additional 3% in a 0.03s
period and finally an additional 2.5% in 250s. The results
of these simulations are visible as the red lines in Figure 5.

As with the closed form expression of the Johnson-Cook
model, the table based method simply interpolates between the
curves it was provided with. Although it is therefore possible
to replicate the isothermal, constant strain rate experimental
data of Figure 2 given the correct table of yield stress versus
plastic strain and plastic strain rate for example, the physically
observed effect of plastic history would not be captured in the
temperature change or strain rate jump tests when using the
table based method.

The state dependent update of the thermally influenced
evolving stress component allows a physically observed
change in the stress state which is not only influenced by

temperature and strain rate. The modelled material response
is clearly also dependent on the plastic history.

IV. MTS USER HARDENING SUBROUTINE

SUBROUTINE UHARD(SYIELD,HARD,EQPLAS,EQPLASRT,
$ TIME,DTIME,TEMP,DTEMP,NOEL,NPT,LAYER,KSPT,
$ KSTEP,KINC,CMNAME,NSTATV,STATEV,NUMFIELDV,
$ PREDEF,DPRED,NUMPROPS,PROPS)

C
IMPLICIT REAL*8(A-H,O-Z)
PARAMETER (NPRECD=2)

C
CHARACTER*80 CMNAME
DIMENSION HARD(3),STATEV(NSTATV),TIME(*),

$ PREDEF(NUMFIELDV),DPRED(*),PROPS(*)
C

ZMU0 = PROPS(1)
D0 = PROPS(2)
TEMP0 = PROPS(3)
SA = PROPS(4)
SI = PROPS(5)
SE0 = PROPS(6)
SES0 = PROPS(7)
TH0 = PROPS(8)
ZKB3 = PROPS(9)
ALPHA = PROPS(10)
G0I = PROPS(11)
G0E = PROPS(12)
G0ES = PROPS(13)
ER0 = PROPS(14)
ER0ES = PROPS(15)
QI = PROPS(16)
PI = PROPS(17)
QE = PROPS(18)
PE = PROPS(19)
T_A = TEMP
T_B = TEMP+DTEMP

C
C INITIALIZE EVOLVING STATE VARIABLE
C

IF(KSTEP.EQ.1.AND.KINC.EQ.1)THEN
STATEV(1) = SE0
STATEV(6) = 1.E-8

ENDIF
C
C CHECK IF THE ANALYSIS HAS MOVED ON TO NEW STEP
C AND INITIALIZE COUNTER STATEV(7)
C

IF(STATEV(8).LT.KSTEP)THEN
STATEV(8) = KSTEP
STATEV(7) = 0.

ENDIF
C
C SET EVOLVING STATE VARIABLE FROM UPDATE
C EVALUATED IN PREVIOUS INCREMENT
C

IF(STATEV(7).LT.KINC)THEN
STATEV(1) = STATEV(1)+STATEV(2)
STATEV(3) = STATEV(4)
STATEV(5) = STATEV(6)
STATEV(7) = KINC

ENDIF
C
C SET VALUE OF SE AT BEGINNIG OF INCREMENT FROM
C STATE VARIABLE AT END OF PREVIOUS
C

SE = STATEV(1)
C
C DETERMINE INITIAL INCREMENT TEMPERATURE EFFECT
C ON VARIABLE ZMU
C

IF(T_A.EQ.0.D0)THEN
C
C USE THRESHOLD VALUE. (SCALE FACTOR = 1)
C

ZMU_A = ZMU0
ELSE

C
C DETERMINE ZMU USING TEMPERATURE SCALE FACTOR
C

ZMU_A = ZMU0-D0/(EXP(TEMP0/T_A)-1.D0)
ENDIF

C
C DETERMINE FINAL INCREMENT TEMPERATURE EFFECT
C ON VARIABLE ZMU
C

IF(T_B.EQ.0.D0)THEN
C
C USE THRESHOLD VALUE. (SCALE FACTOR = 1)
C

ZMU_B = ZMU0
ELSE

C
C DETERMINE ZMU USING TEMPERATURE SCALE FACTOR
C

ZMU_B = ZMU0-D0/(EXP(TEMP0/T_B)-1.D0)
ENDIF

C
C DETERMINE SCALING FACTORS, YIELD STRESS AND
C STATE VARIABLE EVOLUTION ACCORDING TO
C MECHANICAL THRESHOLD STRESS MATERIAL MODEL:
C CHECK PLASTIC DEFORMATION RATE IS HIGHER
C THAN TOLERANCE
C

IF(EQPLASRT.LE.1.E-8)THEN
RATEB = 1.E-8

ELSE
RATEB = EQPLASRT

ENDIF
C
C DETERMINE CHANGE IN PLASTIC DEFORMATION
C

DEPSP = EQPLAS - STATEV(3)
STATEV(4) = EQPLAS
RATEA = STATEV(5)
STATEV(6) = RATEB

C
C CALCULATE LUMPED EQUATION CONSTANTS
C

SFE0 = ZKB3*T_B/(G0E*ZMU_B)
SFE1 = SFE0*LOG(ER0/RATEB)
SFE2 = 1.D0-(SFE1**(1.D0/QE))
SFI0 = ZKB3*T_B/(G0I*ZMU_B)
SFI1 = SFI0*LOG(ER0/RATEB)
SFI2 = 1.D0-(SFI1**(1.D0/QI))
ZKESA = ZKB3*T_A/(G0ES*ZMU_A)
ZKESB = ZKB3*T_B/(G0ES*ZMU_B)

C
C CALCULATE SCALE FUNCTIONS
C

SFE = SFE2**(1.D0/PE)
SFI = SFI2**(1.D0/PI)
SESA = SES0*((RATEA/ER0ES)**ZKESA)
SESB = SES0*((RATEB/ER0ES)**ZKESB)

C
C DETERMINE EVOLUTION AND STRAIN
C

SE_B = SE; FS = 1.;COUNTER = 0
HARD_A = TH0*(1.-(TANH(ALPHA*SE/SESA)

. /TANH(ALPHA)))
C
C NEWTON-RAPHSON UP TO CONVERGENCE OR
C 100 ITERATIONS
C

DO WHILE ((ABS(FS).GT.1.E-8).AND.

. (COUNTER.LT.100))
COUNTER = COUNTER + 1
HARD_B = TH0*(1.0-(TANH(ALPHA*SE_B/SESB)

. /TANH(ALPHA)))
DFS0 = DEPSP*TH0*ALPHA/(TANH(ALPHA)

. *SESB*COSH(ALPHA*SE_B/SESB)**2)
FS = SE_B-SE-DEPSP*(HARD_A+HARD_B)/2.
DFS = 1.+DFS0/2.
SE_B = SE_B - FS/DFS

END DO
HARD_B = TH0*(1.0-(TANH(ALPHA*SE_B/SESB)
. /TANH(ALPHA)))
SE_UPD = SE_B-SE

C
C DETERMINE SYIELD CONSIDERING THE UPDATE ON SE
C

SYIELD = SA+(SFE*SE_B+SFI*SI)*(ZMU_B/ZMU0)
C
C SAVE CURRENT VALUE OF STATE VARIABLE UPDATE IN
C STATEV(2)
C

STATEV(2) = SE_UPD
C
C DERIVATIVES OF SYIELD W.R.T D_EPSP, D_EPSRATE
C AND D_TEMPERATURE
C

DSFE0 = (SFE2**(1.D0/PE - 1.D0))/(PE*QE*RATEB)
DSFE = SFE0*(SFE1**(1.D0/QE - 1.D0))*DSFE0
DSFI0 = (SFI2**(1.D0/PI - 1.D0))/(PI*QI*RATEB)
DSFI = SFI0*(SFI1**(1.D0/QI - 1.D0))*DSFI0

C
C D(SES)/D_ERATE
C

DSES = ZKESB*SES0*((RATEB/ER0ES)**(ZKESB-
. 1.D0))/ER0ES
DFS0 = DEPSP*TH0*ALPHA/(2*TANH(ALPHA)
. *SESB*COSH(ALPHA*SE_B/SESB)**2)
DSE = DFS0*SE_B*DSES/(SESB*(1.+DFS0))

C
C HARD(1) = D(SYIELD)/D(EQPLAS)
C

HARD(1) = SFE*ZMU_B*(HARD_A+HARD_B)/(2.*ZMU0
. *(1+DFS0))

C
C HARD(2) = D(SYIELD)/D(EQPLASRT)
C

HARD(2) = ZMU_B*((DSFI*SI)+(DSFE*SE_B)
. +(SFE*DSE))/ZMU0
RETURN
END

V. CONCLUSION

In this paper, the Mechanical Threshold Stress model was
implemented into an Abaqus UHARD user subroutine. This
user subroutine is also provided in the paper. The use of state
dependent variables is explained and analytical partial deriva-
tives are provided so that the solution is obtained quadratically.

In the examples on real experimental data, the benefit of
having a physical, state variable based plasticity model instead
of a closed form expression or table based plasticity definition
is evident. In applications where the evolution of the yield
stress is required as a function of the plastic history, the MTS
plasticity model included in the paper can be used with little
additional effort.

